UNITED STATES

DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

A COMPUTER PROGRAM FOR CALCULATING RELATIVE-TRANSMISSIVITY INPUT ARRAYS TO AID MODEL CALIBRATION

By Emanuel Weiss

Open-File Report 82-447

Denver, Colorado

1982

UNITED STATES DEPARTMENT OF THE INTERIOR

JAMES G. WATT, Secretary

GEOLOGICAL SURVEY

Dallas L. Peck, Director

For additional information write to:

U.S. Geological Survey
Water Resources Division
Mail Stop 418, Box 25046
Denver Federal Center
Denver, Colorado 80225

For purchase write to:

Open File Services Section
U.S. Geological Survey
Box 25425
Denver Federal Center
Denver, Colorado 80225
(303) 234-5888; FTS 234-5888

CONVERSION FACTORS

To convert inch-pound unit	Multiply by	To obtain metric unit
foot (ft)	3.048×10^{-1}	Meter (m)
pound per square inch	6.895×10^{3}	Newton per square meter (N/m^2)
atmosphere	1.013×10^{5}	Newtons per square meter (N/m^2)
millidarcy	9.870×10^{-10}	Square meter (m^2)
centipoise	1.000×10^{3}	Pascal second (Nt/m^2)

$$°C = 5/9(°F-32)$$

CONTENTS

ILLUSTRATIONS

A COMPUTER PROGRAM FOR CALCULATING RELATIVE-TRANSMISSIVITY INPUT ARRAYS TO AID MODEL CALIBRATION

By Emanuel Weiss

ABSTRACT

A program is documented that calculates a transmissivity distribution for input to a digital ground-water flow model. Factors that are taken into account in the calculation are: aquifer thickness, ground-water viscosity and its dependence on temperature and dissolved solids, and permeability and its dependence on overburden pressure. Other factors affecting ground-water flow are indicated. With small changes in the program code, leakance also could be calculated. The purpose of these calculations is to provide a physical basis for efficient calibration, and to extend rational transmissivity trends into areas where model calibration is insensitive to transmissivity values.

INTRODUCTION

If the average values of ground-water parameters were known for each nodal area of a digital model, the prediction of stress patterns would be the first effort in modeling. In practice, the data on which most ground-water flow models need to be based commonly are few. If this is the case, calibration of the model against historic potentiometric records usually is attempted. In some cases, even adequate historic records are lacking, and a steady-state calibration against one potentiometric surface is made. (Research is continuing on the question of what errors are introduced when a steady-state approach is used for model calibration for a system that is undergoing change.)

The usual method of calibration is a repetitive procedure with trial changes being made in ground-water parameters and the simulated results being compared with field observations. Parameter changes can be made on a trial-and-error basis, with the experience and skill of the modeler playing an important role, or they can be made by a performance analyzer that embodies formal optimization techniques (Neuman, 1973). The discussion in this report is addressed primarily to the former procedure. A Fortran program is documented that calculates a relative transmissivity for each model node by taking into account the dependence of ground-water viscosity on temperature and dissolved-solids concentration and the dependence of aquifer permeability on overburden pressure.

THE NATURE OF TRANSMISSIVITY AND LEAKANCE

In quasi-three-dimensional ground-water models, two of the most important input parameters are aquifer transmissivity and confining-bed leakance. Both contain hydraulic-conductivity factors:

$$T = Kd \tag{1}$$

where

- T is aquifer transmissivity along the direction of K, (L^2T^{-1});
- K is hydraulic conductivity along the aquifer-bedding plane, (LT^{-1});
- d is thickness of the aquifer, (L); and

$$L = K'/d' \tag{2}$$

where

- L is confining-bed leakance along the direction of K, (T^{-1});
- K' is confining-bed hydraulic conductivity in a direction perpendicular to the confining bed, (LT^{-1}); and
- d' is thickness of the confining bed, (L).

Transmissivity and leakance are coefficients in the ground-water flow equations that comprise a quasi-three-dimensional model (Trescott, 1976). As such, they are convenient parameters for model simulations. More basic parameters are factors of each; in particular, hydraulic conductivity contains two parameters that characterize the ground-water and one that characterizes the porous medium:

$$K = \rho g\ k/\mu \tag{3}$$

where

- ρ is density of water present in the rock matrix, (ML^{-3});
- g is acceleration of gravity, (LT^{-2});
- k is intrinsic permeability; sometimes called permeability, specific permeability, or intrinsic hydraulic conductivity, (L^2); and
- μ is dynamic viscosity of water present in the rock matrix, $(ML^{-1}T^{-1})$.

Hydraulic conductivity is associated with freshwater or constant-density ground-water problems. In these problems, usually, only variations of permeability (k) are considered significant. In regions where ground-water pressures, temperatures, or dissolved-solids concentrations have large variations, these variations affect hydraulic conductivity by affecting ground-water viscosity (fig. 1) (Matthews and Russell, 1967) and density. Changes in hydraulic conductivity due to viscosity changes are greater than changes in hydraulic conductivity due to density changes, but density changes may be significant, because the density of natural waters can range from 0.95 to more than 1.20 grams per milliliter (Potter and Brown, 1977).



Figure 1.--Viscosity as a function of temperature, pressure, and dissolved sodium chloride (modified from Matthews and Russell, 1967).

Besides the effects on hydraulic conductivity, the fact that significant changes in density occur affects Darcy's law. Specific discharge is no longer proportional to the gradient of hydraulic head. Furthermore, where significant density changes depend on ground-water pressure, temperature, and dissolved-solids concentrations, variable density flow cannot be described in terms of a potential (Bear, 1972). Because Darcy's law can no longer be written in terms of a potential, the differential equation that results from the combination of Darcy's law and the continuity equation is changed. In making the resulting differential equation analogous to the constant density case Weiss (unpublished data, 1981) finds it convenient to introduce a hybrid hydraulic conductivity:

$$K_h = \rho_p gk/\mu \qquad (4)$$

where ρ_p is the concentration of pure water per unit volume of ground water, ML^{-3}. For problems with dissolved electrolytes flowing through coarse-grained clayey soils, other factors enter into the ground-water flow description (Scheidegger, 1974).

Density and Viscosity Determination

Usually, direct determinations of ground-water density and viscosity are unavailable to model studies. The density and viscosity of brines can be approximated by those of pure sodium chloride solutions of the same ionic strength (Collins, 1975). Once this assumption is made, pure-water concentration and ground-water density can be approximated from tables (Potter and Brown, 1977). A program, based on these tables, that calculates ground-water density and pure-water concentration, has been developed (Weiss, unpublished data, 1981). This program could be combined with the program discussed in this paper to calculate a relative transmissivity dependent on ground-water density or a relative hybrid transmissivity. Throughout the following discussion, it is assumed that model calibration is not sensitive to density-caused changes in transmissivity or leakance distributions.

Viscosity can be approximated from the literature or figure 1 (Matthews and Russell, 1967). An approximate empirical relationship that ignores pressure dependence was developed by the author to fit the graph (fig. 1):

$$\mu = (38.3/T^{\frac{1}{2}} - 14.6/T^{\frac{1}{4}} + 1.48)(1 + DS/300) \qquad (5)$$

where

- μ is viscosity, in centipoise;
- T is temperature, in degrees Fahrenheit; and
- DS is dissolved-solids concentration, in grams per liter.

This approximation's largest error is near 55° Fahrenheit, where it is inaccurate by nearly 10 percent. An example of the use of temperature dependence of viscosity in a model calibration is given by Konikow, 1976.

Permeability

On the basis of data compiled in many textbooks, primary permeability (even of a specific lithology) may vary over several orders of magnitude (Freeze and Cherry, 1979). In addition, consolidated rock commonly has large secondary or fracture permeability. Thus, it is almost impossible to estimate permeability from lithology with the same certainty as viscosity and density without additional information. Usually, the modeler has specific-capacity data, electric logs, aquifer tests, drill-stem tests, or other data to decrease uncertainty of the initial permeability estimate. If no information exists and if recharge or flow estimates can be made, then these estimates can be used in a few trial simulations to decrease permeability uncertainty (Konikow, 1976).

Calibration of the model may improve if permeability is decreased as overburden pressure increases. Here, overburden pressure is the same as effective stress which is used in aquifer mechanics (Poland, 1972). Effective stress (overburden pressure) is the difference between the pressure exerted downward by the sediments and liquids above a point within the saturated deposit and the liquid pressure at the same point. Thus, effective stress and overburden pressure are the intergranular pressures within the aquifer.

For sandstone, some relative-permeability curves, as a function of effective stress, are shown in figure 2. The permeability-overburden relationship in the Fortran code in the last section of this paper is a concatenation of line segments that plots between curves 2 and 3 in figure 2. This choice corresponds to an unpressurized permeability of 40 millidarcies or greater. For unpressurized permeability about 4 millidarcies, a choice between curves 1 and 4 is appropriate. To calculate overburden pressure, in pounds per square inch, from overburden thickness, in feet, multiplication by a factor ranging from 0.5 to 0.25 gives a good approximation (Core Laboratories, Inc., 1974). In the following program, a factor of 0.5 is used.

Although the present discussion is specifically a relative transmissivity calculation for sandstone, the relative leakance for a confining bed easily could be calculated using the same principles. A comparison of equations (1) and (2) shows the principal differences. A laboratory determination of vertical permeability as a function of overburden pressure for clay (kaolinite) shows that overburden can change vertical permeability by several orders of magnitude (fig. 3). This could be used in a relative leakance calculation for a clay confining bed.

There is some ambiguity caused by the hysteresis of clays. Whether one is on the rebound portion or the consolidation portion of the curve can cause an order of magnitude difference in the value of permeability associated with an overburden pressure. Another factor to consider is the effect of dewatering. For a discussion of how this can affect aquifer and confining-bed permeability, see Helm, 1975 and 1976. Another factor to consider is the presence of residual, anomalously-high fluid pressures within low-permeability material. These pressures can cause the expected effective stress on the granular matrix to be less than expected (Rieke, 1974). Each of these factors may be used as justification for changes to the calculated relative-transmissivity arrays rather than incorporation into a computer code.



Figure 2.--Change in permeability of sandstone with overburden pressure (modified from Fatt and Davis, 1952).



Figure 3.--Change in permeability of kaolinite with overburden pressure (modified from Olsen, 1972).

COMPUTER CODE

Data-Deck Instructions

The following tabulation lists the data input required for this example and their corresponding formats.

CARD	COLUMN	FORMAT	VARIABLE	DEFINITION
1	1-10	I10	IØ	Number of rows.
	11-20	I10	JØ	Number of columns.
2	1-80	20F4.0	VK	Values of permeability type, in any consistent unit (20 values possible). Type 1 is first; type 2 is second, and so forth.

All of the following input data require a parameter card, either preceding an array or as the sole input for that data.

Every parameter card	1-10	G10.0	FAC	If IVAR = 0, FAC is the value assigned to every element of the matrix for this layer. If IVAR = 1, FAC is the multiplication factor for the following set of data cards for this layer.
	11-20	G10.0	IVAR	0--If no data cards are to be read in for this layer. 1--If data cards for this layer follow.
	21-30	G10.0	IPRN	0--If input data for this layer are to be printed. 1--If input data for this layer are not to be printed.

When arrays are included, start each row on a new card.

DATA SET	COLUMN	FORMAT	VARIABLE	DEFINITION
1	1-80	20F4.0	TPK	Permeability type array, 1 to 20 types. Numbers used to designate type are 1 to 20 (unitless).
2	1-80	20F4.0	T1	Temperature array, in degrees Fahrenheit.
3	1-80	20F4.0	THK	Aquifer thickness, in feet.
4	1-80	20F4.0	TELEV	Topographic elevation, in feet.
5	1-80	20F4.0	AELEV	Aquifer elevation, in feet.
6	1-80	20F4.0	DS	Dissolved-solids concentration, in grams per liter.

Example of a Relative-Transmissivity Calculation

It is the purpose of this example to illustrate the data input and results of a relative-transmissivity calculation for a hypothetical dipping-sandstone aquifer of constant thickness. Temperature, dissolved-solids concentration, permeability, and overburden pressure increase in the direction of dip.

The program is written so that 20 permeability types may be entered for each aquifer. A permeability type may correspond to an area of similar lithology. Once these areas are identified, each node in an area is assigned the same value of permeability by the program. This programming avoids having to repunch more than one number, when the value of any permeability type is changed for calibrations runs.

It can be seen from the output of the calculated relative transmissivity that the largest value is 9,998. This is true for every calculation. Furthermore, the relative transmissivity can be put directly into group III, data set 4, of the program in Trescott, 1976. In this program, the number entered on the factor-card multiplies each number of the relative-transmissivity array to calculate input transmissivity.

<u>Data Input for Dipping-Sandstone-Aquifer Example</u>

IØ, JØ			5			10					
VK	0	1-3		1-2	1-1	1	0				
			1			1					
TPK	1	1		1	1	1	1	1	1	1	1
	1	2		3	4	5	5	5	5	5	1
	1	2		3	4	5	5	5	5	5	1
	1	2		3	4	5	5	5	5	5	1
	1	1		1	1	1	1	1	1	1	1
			1			1					
T1	0	0		0	0	0	0	0	0	0	0
	0	50		80	110	140	170	200	230	260	0
	0	50		80	110	140	170	200	230	260	0
	0	50		80	110	140	170	200	230	260	0
	0	0		0	0	0	0	0	0	0	0
THK			100								
TELEV			2000								
			100			1					
AELEV	0	0		0	0	0	0	0	0	0	0
	0	20		10	0	-10	-20	-30	-40	-50	0
	0	20		10	0	-10	-20	-30	-40	-50	0
	0	20		10	0	-10	-20	-30	-40	-50	0
	0	0		0	0	0	0	0	0	0	0
			1			1					
DS	0	0		0	0	0	0	0	0	0	0
	0	5		10	15	20	30	50	100	400	0
	0	5		10	15	20	30	50	100	400	0
	0	5		10	15	20	30	50	100	400	0
	0	0		0	0	0	0	0	0	0	0
//											
Column numbers on computer card	4	8	10	12	16	20	24	28	32	36	40

Output for Dipping-Sandstone-Aquifer Example

INTRINSIC-K TYPE NO.   MATRIX, LAYER 1

1	1.0	1.0	1.0	1.0	1.0	1.0	1.0	1.0	1.0	1.0
2	1.0	2.0	3.0	4.0	5.0	5.0	5.0	5.0	5.0	1.0
3	1.0	2.0	3.0	4.0	5.0	5.0	5.0	5.0	5.0	1.0
4	1.0	2.0	3.0	4.0	5.0	5.0	5.0	5.0	5.0	1.0
5	1.0	1.0	1.0	1.0	1.0	1.0	1.0	1.0	1.0	1.0

AQUIFER TEMP DATA   MATRIX, LAYER 1

1	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
2	0.0	0.500E 02	0.800E 02	0.110E 03	0.140E 03	0.170E 03	0.200E 03	0.230E 03	0.260E 03	0.0
3	0.0	0.500E 02	0.800E 02	0.110E 03	0.140E 03	0.170E 03	0.200E 03	0.230E 03	0.260E 03	0.0
4	0.0	0.500E 02	0.800E 02	0.110E 03	0.140E 03	0.170E 03	0.200E 03	0.230E 03	0.260E 03	0.0
5	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0

AQUIFER THICKNESS = 100.0000 FOR LAYER 1

TOPO-ELEVATION IN FT = 2000.000 FOR LAYER 1

AQUIFER ELEVATION IN FT MATRIX, LAYER 1

1	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
2	0.0	0.200E 04	0.100E 04	0.0	-0.100E 04	-0.200E 04	-0.300E 04	-0.400E 04	-0.500E 04	0.0
3	0.0	0.200E 04	0.100E 04	0.0	-0.100E 04	-0.200E 04	-0.300E 04	-0.400E 04	-0.500E 04	0.0
4	0.0	0.200E 04	0.100E 04	0.0	-0.100E 04	-0.200E 04	-0.300E 04	-0.400E 04	-0.500E 04	0.0
5	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0

*** DS (IN G/LITER)**** MATRIX, LAYER 1

1	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0
2	0.0	0.500E 01	0.100E 02	0.150E 02	0.200E 02	0.300E 02	0.500E 02	0.100E 03	0.400E 03	0.0
3	0.0	0.500E 01	0.100E 02	0.150E 02	0.200E 02	0.300E 02	0.500E 02	0.100E 03	0.400E 03	0.0
4	0.0	0.500E 01	0.100E 02	0.150E 02	0.200E 02	0.300E 02	0.500E 02	0.100E 03	0.400E 03	0.0
5	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0

MAXIMUM & MINIMUM TRANS(I,J)= 206.9028 0.0

RELATIVE TRANSMISSIVITY
****

0	0	0	0	0	0	0	0	0	0
0	3	49	641770490529998997	66564					0
0	3	49	641770490529998997	66564					0
0	3	49	641770490529998997	66564					0
0	0	0	0	0	0	0	0	0	0

AQUIFER THICKNESS
****

100.	100.	100.	100.	100.	100.	100.	100.	100.	100.
100.	100.	100.	100.	100.	100.	100.	100.	100.	100.
100.	100.	100.	100.	100.	100.	100.	100.	100.	100.
100.	100.	100.	100.	100.	100.	100.	100.	100.	100.
100.	100.	100.	100.	100.	100.	100.	100.	100.	100.

(INTRINSIC PERMEABILITY)

K-VALUES FOR LAYER
****

0.	0.	0.	0.	0.	0.	0.	0.	0.	0.
0.	0.	0.	0.	1.	1.	1.	1.	1.	0.
0.	0.	0.	0.	1.	1.	1.	1.	1.	0.
0.	0.	0.	0.	1.	1.	1.	1.	1.	0.
0.	0.	0.	0.	0.	0.	0.	0.	0.	0.

OVERBRDN DELTA
****

0.87	0.87	0.87	0.87	0.87	0.87	0.87	0.87	0.87	0.87
0.87	1.00	0.93	0.87	0.80	0.77	0.73	0.70	0.69	0.87
0.87	1.00	0.93	0.87	0.80	0.77	0.73	0.70	0.69	0.87
0.87	1.00	0.93	0.87	0.80	0.77	0.73	0.70	0.69	0.87
0.87	0.87	0.87	0.87	0.87	0.87	0.87	0.87	0.87	0.87

VISCOSITY
****

1.18	1.18	1.18	1.18	1.18	1.18	1.18	1.18	1.18	1.18
1.18	1.43	0.91	0.65	0.50	0.41	0.35	0.34	0.51	1.18
1.18	1.43	0.91	0.65	0.50	0.41	0.35	0.34	0.51	1.18
1.18	1.43	0.91	0.65	0.50	0.41	0.35	0.34	0.51	1.18
1.18	1.18	1.18	1.18	1.18	1.18	1.18	1.18	1.18	1.18

* RAN TO END OF PROGRAM **

Definitions of Some Program Variables

AELEV	Elevation of the aquifer middle, in feet above sea level.
DELTA	Ratio of permeability with overburden pressure to permeability without overburden pressure.
IØ, JØ	Number of rows and columns in model grid.
KK	Array of relative transmissivity with maximum member equal to 9,998.
PSI	Overburden pressure, in pounds per square inch.
T1	Aquifer temperature, in degrees Fahrenheit.
DS	Dissolved-solids concentration of water in aquifer, in grams per liter.
DSMU	A factor expressing viscosity's dependence on dissolved-solids concentration.
TELEV	Elevation of land surface, in feet above sea level.
TEMP	Intermediate variable to which temperature is assigned.
THK	Aquifer thickness array.
TPK	Aquifer permeability type array.
TRANS	Calculated relative transmissivity.
VK	Estimated values of each permeability type without overburden.
XK	Array of estimated permeability without overburden.
XMU	A factor expressing viscosity's dependence on temperature.
XMAX, XMIN	Maximum and minimum values of calculated transmissivity.
ZMU	Calculated viscosity, in centipoises.

Fortran Program to Calculate Relative Transmissivity

The following Fortran program for calculating relative transmissivity was written for a specific model study and contains code peculiar to that study or data used in that study. Some of these peculiarities are:

1. The arrays are dimensioned 21 x 26.
2. The decrease in permeability caused by overburden pressure is calculated for sandstone.
3. Aquifer transmissivity is calculated, but simple changes can convert the program to a leakance calculation for a confining bed.
4. The largest value of relative transmissivity is always 9,998.
5. For a ground-water temperature of less than 40° Fahrenheit, transmissivity and viscosity are set equal to zero. This allows the user to identify input errors in a temperature array.
6. A factor of 0.5 is used to determine overburden pressure; it can be decreased to 0.25.

None of these peculiarities are difficult to change to suit an individual study.

The program was run on the U.S. Geological Survey's Amdahl 470 V/7* computer in Reston, Virginia, with an IBM System 360/370, Fortran IV, H-extended compiler. Typical central processing unit time is 0.4 second.

---

*Any use of trade names is for descriptive purposes only and does not imply endorsement by the U.S. Geological Survey.

```
C*********************************************************************
C     THIS PROGRAM CALCULATES A RELATIVE AQUIFER TRANSMISSIVITY.       *
C      FACTORS CONSIDERED: VISCOSITY AND ITS DEPENDENCE ON  DS &      *
C                          TEMPERATURE, OVERBURDEN PRESSURE, THKNES   *
C                     AND OPTION TO INPUT UNPRESSURIZED INTRINSIC     *
C                           HYDRAULIC CONDUCTIVITY.....               *
C ******************************************************************  *
      INTEGER BLK
      DIMENSION TPK(21,26), T1(21,26), THK(21,26), TRANS(21,26), INFT(2,
     12), IOFT(9,4), IN(6), DUM(3), TELEV(21,26), AELEV(21,26), KK(21,26
     2),  DS(21,26), VK(20), XK(21,26), DELTA(21,26), ZMU(21,26)
      COMMON /MISC/ I0,J0
      DATA BLK/' '/
      DATA INFT/4H(20F,4H4.0),4H(8F1,4H0.4)/
      DATA IOFT/'(1H0','I2,','2X,2','0F6.','1/(5','X,20','F6.1',')) ',
     1'    ','(1H0','I5,','14F9','.5/(','6X,1','4F9.','5)) ','    ','
     2 ','(1H0','I5,','10E1','2.5/','(6X,','10E1','2.5)',')   ','    '
     3,'(1H0','I5,','10E1','1.3/','(6X,','10E1','1.3)',')   ','    '/
C********
      IRN=1
      XMAX=0.0
      XMIN=9999.0
      DO 110 I=1,6
  110 IN(I)=BLK
C *******************************************************************
C       READ IN UNPRESSURIZED INTRINSIC HYDRAULIC CONDUCTIVITY VALUE
C       ASSOCIATED W EACH TYPE DEFINED IN NEXT READ-IN STATEMENT
C       ( IN MILLIDARCIES, TYPICAL INPUT 3+4 FOR GRAVEL, 9 FOR
C          POOR AQUIFERS.)                                          *
C *******************************************************************
      READ (5,210) I0,J0
C-----READ IN INTRINSIC-K VALUES. VK(1) ASSOCIATED W TYPE 1 (TPK(1))
      READ (5,220) (VK(J),J=1,20)
C ---- READ IN INTRINSIC-K TYPE
      CALL ARRAY(TPK,INFT(1,1),IOFT(1,1),'INTRINSIC-K TYPE NO.',IRN,DUM)
C---- READ IN TEMPERATURE OF AQUIFER
      CALL ARRAY(T1,INFT(1,1),IOFT(1,4),' AQUIFER  TEMP DATA ',IRM,DUM)
C---- READ IN THICKNESS OF AQUIFER
      CALL ARRAY(THK,INFT(1,1),IOFT(1,4),'AQUIFER   THICKNESS',IRN,DUM)
C*******************************************************************
C                    TELEV & AELEV MUST BE IN FEET
C     READ IN TOPOGRAPHIC ELEVATION OF PROJECT AREA IN FT ABOVE MSL.
C*******************************************************************
      CALL ARRAY(TELEV,INFT(1,1),IOFT(1,4),'TOPO-ELEVATION IN FT
     1ABOVE MSL',IRN,DUM)
C---- READ IN  AQUIFER MIDDLE ELEVATION IN FT ABOVE MSL
      CALL ARRAY(AELEV,INFT(1,1),IOFT(1,4),'AQUIFER ELEVATION IN FT',IRN
     1,DUM)
C---- READ IN DISSOLVED SOLIDS OF AQUIFER (IN G/LITER)
      CALL ARRAY (DS,INFT(1,1),IOFT(1,4),'***  DS (IN G/LITER)*******',I
```

```
     1RN,DUM)
      DO 140 I=1,I0
      DO 140 J=1,J0
      XK(I,J)=0.0
      KK(I,J)=0
      ZMU(I,J)=0.0
      DELTA(I,J)=0.0
C**********************************************************************
C     BEGINNING OF VISCOSTY CALCULATION
C**********************************************************************
      IF (THK(I,J).EQ.0) GO TO 130
      IF (T1(I,J).LT.40) GO TO 130
      IF (T1(I,J).NE.0) TEMP=T1(I,J)
      XMU=38.3432/SQRT(TEMP)-14.623/SQRT(SQRT(TEMP))
      YMU=XMU+1.481
      DSMU=1.+(DS(I,J)/300.)
      ZMU(I,J)=YMU*DSMU
      DO 120 K=1,20
      IF (TPK(I,J).EQ.K) XK(I,J)=VK(K)
  120 CONTINUE
C**********************************************************************
C  ARRAY AELEV(I,J) IS ELEV OF MIDDLE OF AQUIFER                      *
C**********************************************************************
      PSI=(TELEV(I,J)-AELEV(I,J))/2.
      IF (PSI.LE.1500) DELTA(I,J)=1.00-0.2*(PSI/1500.)
      IF (PSI.LE.0) DELTA(I,J)=1.0
      IF (PSI.GT.1500.AND.PSI.LE.3000) DELTA(I,J)=0.8-0.1*((PSI-1500.)/15
     100.)
      IF (PSI.GT.3000.AND.PSI.LE.12000) DELTA(I,J)=.7-.2*((PSI-3000.)/90
     100.)
      IF (PSI.GT.12000) DELTA(I,J)=0.45
C**********************************************************************
C     RELATIVE TRANSMISSIVITY IS CALCULATED NEXT                      *
C**********************************************************************
      TRANS(I,J)=XK(I,J)*THK(I,J)*DELTA(I,J)/ZMU(I,J)
      IF (TRANS(I,J).GT.XMAX) XMAX=TRANS(I,J)
      IF (TRANS(I,J).LT.XMIN) XMIN=TRANS(I,J)
      GO TO 140
  130 TRANS(I,J)=0.0
      ZMU(I,J)=0.0
  140 CONTINUE
C*********************************************************************
C       9999./XMAX MAKES THE MAX TRANS(I,J) VALUE 9999.
C*********************************************************************
      WRITE (6,240) XMAX,XMIN
      WRITE (6,230)
      DO 160 I=1,I0
      DO 150 J=1,J0
  150 KK(I,J)=INT(TRANS(I,J)*9999./XMAX)
      WRITE (6,250) (KK(I,J),J=1,J0)
```

```
      WRITE (7,260) (KK(I,J),J=1,J0)
  160 CONTINUE
      WRITE (6,270)
      DO 170 I=1,I0
      WRITE (6,280) (THK(I,J),J=1,J0)
  170 CONTINUE
      WRITE (6,290)
      DO 180 I=1,I0
      WRITE (6,300) (XK(I,J),J=1,J0)
  180 CONTINUE
      WRITE (6,310)
      DO 190 I=1,I0
      WRITE (6,320) (DELTA(I,J),J=1,J0)
  190 CONTINUE
      WRITE (6,330)
      DO 200 I=1,I0
      WRITE (6,340) (ZMU(I,J),J=1,J0)
  200 CONTINUE
      WRITE (6,350)
C     ********************
      STOP
C
C
  210 FORMAT (2I10)
  220 FORMAT (20F4.0)
  230 FORMAT (1H1,15X,'RELATIVE TRANSMISSIVITY',/,16X,'**********')
  240 FORMAT (1H ,10X,'MAXIMUM & MINIMUM TRANS(I,J)=',F10.4,3X,F10.4,//)
  250 FORMAT (T2,26I4,/)
  260 FORMAT ((20I4,/),6I4)
  270 FORMAT (1H1,15X,'  AQUIFER   THICKNESS  ',/,16X,'***************')
  280 FORMAT (T2,26F5.0,/)
  290 FORMAT (1H1,15X,'K-VALUES FOR LAYER ',/,16X,'********************')
  300 FORMAT (T2,26F5.0,/)
  310 FORMAT (1H1,15X,'OVERBRDN DELTA',/,16X,'********************')
  320 FORMAT (T2,26F5.2,/)
  330 FORMAT (1H1,15X,'VISCOSITY ',/,16X,'*******************')
  340 FORMAT (T2,26F5.2,/)
  350 FORMAT (//,'** RAN TO END OF PROGRAM **')
      END
```

SUBROUTINE ARRAY(A,INFT,IOFT,IN,IRN,TF)

```
      SUBROUTINE ARRAY(A,INFT,IOFT,IN,IRN,TF)
C     *********************
      INTEGER BLK
      DIMENSION A(21,26), INFT(2,2), IN(6), TF(3), DUM(3), DUM2(4), IOFT
     1(9,4)
      COMMON /MISC/ I0,J0
      DATA BLK/' '/
      K=1
      READ (5,120) FAC,IVAR,IPRN,TF,IRECS,IRECD
      IC=4*IRECS+2*IVAR+IPRN+1
      GO TO (10,10,30,30,60,60), IC
   10 DO 20 I=1,I0
      DO 20 J=1,J0
   20 A(I,J)=FAC
      WRITE (6,100) IN,FAC,K
      GO TO 80
   30 IF (IC.EQ.3) WRITE (6,110) IN,K
      DO 50 I=1,I0
      READ (5,INFT) (A(I,J),J=1,J0)
      DO 40 J=1,J0
   40 A(I,J)=A(I,J)*FAC
   50 IF (IC.EQ.3) WRITE (6,IOFT) I,(A(I,J),J=1,J0)
      GO TO 80
   60 CONTINUE
      IF (IC.EQ.6) GO TO 80
      WRITE (6,110) IN,K
      DO 70 I=1,I0
   70 WRITE (6,IOFT) I,(A(I,J),J=1,J0)
   80 CONTINUE
      IRN=IRN+1
      DO 90 I=1,6
   90 IN(I)=BLK
      RETURN
C
C     ---FORMATS---
C
C
C
  100 FORMAT (1H0,52X,6A4,' =',G15.7,' FOR LAYER',I3)
  110 FORMAT (1H1,45X,6A4,' MATRIX, LAYER',I3/46X,41('-'))
  120 FORMAT (F10.0,2I10,3F10.0,2I10)
      END
```

REFERENCES CITED

Bear, Jacob, 1972, Dynamics of fluids in porous media: New York, American Elsevier Scientific Publishing Co., Inc., p. 159-161.

Collins, A. G., 1975, Geochemistry of oilfield waters: New York, American Elsevier Scientific Publishing Co., Inc., p. 172-173.

Core Laboratories, Inc., 1974, Special core analysis: Dallas, Special Core Analysis Studies Section, p. 138.

Fatt, I., and Davis, D. H., 1952, Reduction in permeability with overburden pressure: Petroleum Transactions, American Institute of Mining and Metallurgical Engineers, v. 195, p. 329.

Freeze, R. A., and Cherry, J. A., 1979, Groundwater: Englewood Cliffs, New Jersey, Prentice-Hall, Inc., p. 29.

Helm, D. C., 1975, One-dimensional simulation of aquifer system compaction near Pixley, California, 1, constant parameters: Water Resources Research, v. 11(3), p. 465-478.

______ 1976, One-dimensional simulation of aquifer system compaction near Pixley, California, 2, stress-dependent parameters: Water Resources Research, v. 12(3), p. 375-391.

Konikow, L. F., 1976, Preliminary digital model of ground-water flow in the Madison Group, Powder River Basin and adjacent areas, Wyoming, Montana, South Dakota, North Dakota, and Nebraska: U.S. Geological Survey Water-Resources Investigations 63-75, 44 p.

Matthews, C. S., and Russell, D. C., 1967, Pressure buildup and flow tests in wells, appendix G: Dallas, Society of Petroleum Engineers of the American Institute of Mining, Metallurgical, and Petroleum Engineers Monograph 1, p. 158.

Neuman, S. P., 1973, Calibration of distributed parameter groundwater flow models viewed as a multiple-objective decision process under uncertainty: Water Resources Research, v. 9, p. 1006-1021.

Olsen, H. W., 1972, Liquid movement through kaolinite under hydraulic, electric, and osmotic gradients: American Association of Petroleum Geologists Bulletin, v. 56, no. 10, p. 2022-2028.

Poland, J. F., Lofgren, B. E. and Riley, F. S., 1972, Glossary of selected terms useful in studies of the mechanics of aquifer systems and land subsidence due to fluid withdrawal: U.S. Geological Survey Water-Supply Paper 2025, 9 p.

Potter, R. W., and Brown, D. L., 1977, The volumetric properties of aqueous sodium chloride solutions from 0° to 500°C at pressures up to 2000 bars based on a regression of available data in the literature: U.S. Geological Survey Bulletin 1421-C, 36 p.

Rieke, H. H., and Chilingarian, G. V., 1974, Compaction of argillaceous sediments: New York Elsevier Scientific Publishing Co., 305 p.

Scheidegger, A. E., 1974, The physics of flow through porous media: Toronto, University of Toronto Press, p. 179-184.

Trescott, P. C., 1976, Documentation of finite-difference model for simulation of three-dimensional ground-water flow: U.S. Geological Survey Open-File Report 75-438, 103 p.

Weiss, Emanuel, 1982, A model for the simulation of flow of variable-density ground water in three dimensions under steady-state conditions: U.S. Geological Survey Open-File Report 82-352, 60 p.